What’s a COS file and how do I make one?








A COS file (Creatures Object Source) is purely a text file with a .COS extension. The contents of the file and its arrangement are what make it special, and allow it to be used with the CyberLife COB Compiler.





A COS file intended to become a COB can have up to 3 distinct parts within it; the Installation Script, the Events Scripts and the Remove Scripts.


The Installation Script will be used to create the physical presence of any objects needed in the world (for example, a seed box) and place it in Albia.


The Event Scripts are all the scripts that are needed to be stored in the scriptorium, and usually correspond to the event scripts needed by the object you created in the install part.


The Remove Scripts are the commands that are needed to remove all presence of the object from the world, including its scripts.





With this in mind, the structure of a COS file can be shown as below:


----------------------------------------------------------------------------------------------------------------------------





*****************EXAMPLE COS FILE – this will make a ball


*** You can use asterisks for marking a line as a comment 


*** and it will be ignored by the COB Compiler











******** Here’s the Installation part – use the token ‘iscr’ to mark this part of the


********the COS file as being for installation.





iscr





***Place all the commands to create the object(s) here, one command per line.


***You can use spacing and tabs to make the code more readable For example,


new: simp bal2 8 0 500 0


setv cls2 2 13 4


setv attr 199


bhvr 1 11





setv size 110


setv accg 5


setv rest 20


setv aero 5





mvto 5300 400


slim


setv grav 1


endm








********Here’s some Event Scripts – the compiler will know that the Installation part has finished by the use of ‘endm’ at the end of the ‘iscr’ section, or by the use of the ‘scrp’ command.





***ball collision


scrp 2 13 4 6


	snde boi2


	setv actv 0


	doif wall eq 3


		doif velx lt 0


			anim [07654321]


		endi


		doif velx gt 0


			anim [12345670]


		endi


	endi


	doif wall eq 0


		anim [56765]


	endi


	doif wall eq 1


		anim [56765]


	endi


	doif grav eq 0


		pose 0


	endi


	stim shou 40 8 0 0 36 50 45 50 42 50 0 0


endm





***ball activate


scrp 2 13 4 1


	stim writ from 20 9 20 0 42 10 36 50 45 100 44 50


	setv var0 posl


	setv var1 posr


	addv var0 var1


	divv var0 2


	targ from


	setv var1 posl


	targ ownr


	doif var1 lt var0


		rndv var2 15 25


		anim [12345670]


	else


		rndv var2 -25 -15


		anim [07654321]


	endi


	setv velx var2


	setv vely -40


	setv actv 0


endm








********Here’s the removal part – these commands will remove the physical presence of the object ********from Albia, but also remove the scripts it injected. Use the token ‘rscr’ to mark this section as ********containing the removal scripts





rscr





enum 2 13 4


	kill targ


next





scrx 2 13 4 6


scrx 2 13 4 1





endm











And that’s it!


The full range of CAOS commands are detailed in a different document, the purpose of this is to show that a COS file is a simple data structure – and can be created with a text editor.


So how can I turn a COS file into a COB I can distribute?








By creating a COS file in the format detailed above you are essentially creating the bare bones of a COB. All that is then needed is to pass it through a COB Compiler (available from http:\\www.creatures2.com) and add the extra files needed for the object to work (images and sounds).





Some points to bear in mind when creating a COS with the intention of making a COB:


Every object type you create needs a unique classifier (Family – Genus – Species) to identify it and it’s event scripts. If an object you create clashes with the classifier of another object then the scriptorium will become corrupted and it is likely that one of the clashing objects will cause errors. To remove this possibility CyberLife will be maintaining a database of object classifiers already in use, from which people can request an unused classifier. This mechanism is not fully implemented yet but in the mean time people can use classifiers in the species range of 20,000 – 65,000 and hope they don’t chose a number someone else has.


All COBs require a remove script – this allows the user to remove the object from their world if they decide they no longer want it there. Remove scripts should also remove any event scripts the COB addded, this will prevent the scriptorium from becoming full of obsolete events.








A note on classifiers:





CyberLife will be maintaining a database of classifiers as a part of the CDN program, this will allow 3rd parties to reserve classifiers for objects they are making and allows them to guarantee that their objects will not clash with anyone else’s who uses the CDN program.


To allow the freedom of not belonging to the CDN there is a range of classifiers that will not be maintained and are essentially ‘use it with care’ objects. When creating objects in this range it is highly advisable for everyone to explicitly state what classifiers they are using (in the documentation for the COB) to help reduce the chance of clashing.





The species aspect of the classifier has a range from 1 to 65,535 – but not all of these are available for use. The table below shows the allocation of classifiers:





1 … 1,999		Reserved for CyberLife Use


2,000 … 19,999	Reserved for CDN allocation


20,000 +		Free for all

